Calculating Incoherent Diffraction MTF

Melvin Friedman, Jay Vizgaitis

US Army RDECOM
Communications-Electronics Research and Development
Night Vision and Electronic Sensors Directorate
Fort Belvoir, VA

ABSTRACT

The incoherent diffraction MTF plays an increasingly important role in the range performance of imaging systems as the wavelength increases and the optical aperture decreases. Accordingly, all NVESD imager models have equations that describe the incoherent diffraction MTF of a circular entrance pupil. NVThermIP, a program which models thermal imager range performance, has built in equations which analytically model the incoherent diffraction MTF of a circular entrance pupil and has a capability to input a table that describes the MTF of other apertures. These can be calculated using CODE V, which can numerically calculate the incoherent diffraction MTF in the vertical or horizontal direction for an arbitrary aperture. However, we are not aware of any program that takes as input a description of the entrance pupil and analytically outputs equations that describe the incoherent diffraction MTF. This work explores the effectiveness of *Mathematica* to analytically and numerically calculate the incoherent diffraction MTF for an arbitrary aperture. In this work, *Mathematica* is used to analytically and numerically calculate the incoherent diffraction MTF for a variety of apertures and the results are compared with CODE V calculations.

Keyword list. Incoherent, Diffraction, Optical diffraction, Modulation Transfer Function, Optical Transfer Function, MTF, OTF, Mathematica, Analytical calculation

1. INTRODUCTION

Objectives of Research. NVThermIP is a well-known NVESD developed program[1] that takes as input sensor, atmospheric, and target parameters, and outputs the range performance of thermal imaging systems. One of the sensor input parameters needed by NVThermIP is the incoherent diffraction modulation transfer function (MTF). The analytic expression for the incoherent diffraction MTF of an unobstructed circular aperture is built into NVThermIP. However, if the sensor entrance pupil has a shape other than circular then the incoherent diffraction MTF is input to NVThermIP as a table. The optical MTF table is normally calculated by the programs CODE V or ZEMAX or measured in the lab as an effective optical MTF. One objective of this work is to enable the calculation of an incoherent diffraction MTF table suitable for input into NVThermIP by using the *Mathematica* program A second objective is to use *Mathematica* to produce exact analytical formulae that can be built into NVThermIP and thus make NVThermIP directly applicable to a wider range of optical designs without the need to input incoherent diffraction MTF tables appropriate to that sensor.

Objectives of Paper. The objectives of the paper are: 1) exhibit the subroutines that analytically and numerically calculate incoherent diffraction MTF, 2) demonstrate the use of these subroutines, 3) demonstrate that the routines reproduce well-known analytical results and can produce hitherto unknown

Infrared Imaging Systems: Design, Analysis, Modeling, and Testing XIX,
edited by Gerald C. Holst, Proc. of SPIE Vol. 6941,
69410M, (2008) · 0277-786X/08/$18 · doi: 10.1117/12.779230

analytical results and 4) demonstrate the validity of the exhibited code by comparison with CODE V calculations.

Software Description. CODE V and ZEMAX are optical ray tracing codes that have an ability to numerically calculate the incoherent diffraction MTF associated with a variety of entrance pupils[2,3]. *Mathematica* is a general programming language used mainly for mathematical calculation which has outstanding analytical, numerical, and graphical capabilities[4].

Other Work. Several books[5-15, 22,23] and research papers[17-21] either discuss how to calculate incoherent diffraction MTF or give formulas for different geometries. An unpublished report[16] was most useful in verifying and validating the *Mathematica* codes presented here.

Outline. Section 2 describes the methods used here to calculate incoherent diffraction MTF. Analytical incoherent diffraction MTF formulas are given in Section 3. Section 4 describes how the *Mathematica* codes developed here were verified. Conclusions are given in Section 5 and the *Mathematica* code used is given in the Appendix. The reader is encouraged to first peruse the Appendix with the objective of understanding how to use the functions defined there.

2. THEORY



Figure 1. Defining the autocorrelation of the entrance pupil

The theory of incoherent diffraction is discussed elsewhere[5-15, 22, 23]. Here we only describe relationships utilized in the Appendix. The top part of Fig. 1 defines the optical entrance pupil shape. Here an elliptical shaped pupil is shown but the entrance pupil shape may be arbitrary and include obstructions. The bottom part of Fig. 1 shows the original entrance pupil and the entrance pupil displaced by the distance s. Let A_0 and A_{OL} denote the area of the entrance pupil and the overlap area when the pupil is displaced by the distance s. Here it is assumed that the units of s are mm and the units of A_0 and A_{OL} are mm^2. The overlap area is a function of the displacement *s*. Let λ denote the wavelength of the incident light in units of micrometers and fl denote the focal length of the lens in mm. Then the optical transfer function is given by

$$\mathrm{MTF}(f) = \frac{A_{\mathrm{OL}}(s)}{A_0}\bigg|_{s\to 10^{-3}\,\lambda\,\mathrm{fl}\,f} \quad \text{where spatial frequency } f \text{ is in cycles/mm} \tag{1a}$$

$$\mathrm{MTF}(f) = \frac{A_{\mathrm{OL}}(s)}{A_0}\bigg|_{s\to \lambda\, f} \quad \text{where spatial frequency } f \text{ is in cycles/mr} \tag{1b}$$

In equation (1a) the factor 10^{-3} allows the frequency *f* to be in cycles/mm when the units of *s* and fl are mm and the units of λ are micrometers.

Although Fig. 1 shows the displacement s in the positive x-direction the displacement could just as well have been taken in the y-direction or in any intermediate direction. With s taken to be in the x-direction the frequency in equation 1 is a frequency in the x-direction and is denoted by f_x ; with s taken to be in the y-direction the frequency in equation 1 is a frequency in the y-direction and is denoted by f_y.

The calculation of $A_{OL}(s)$ can be difficult to do analytically by hand for two reasons: 1) For obscured apertures the integral limits change with s and the size of the obscuration; 2) Even when the limits are known the integrals are sometimes difficult to evaluate analytically. A strength of *Mathematica*, is that it allows $A_{OL}(s)$ to be calculated without the user having to explicitly specify the limits of integration. As illustrated in the Appendix this is done by using the *Mathematica* defined functions **Boole** and **Integrate**.

The above equations are useful if the MTF is to be expressed directly in cycles/mm or cycles/mr. Sometimes it is convenient to express the MTF in terms of a normalized frequency f_n. As s increases, for all s values larger then s_{max} the overlap area A_{OL} is zero. Typically the s_{max} value in the x-direction is different from the s_{max} value in the y-direction. An optical cutoff frequency f_{oco} in the x and y directions are defined in terms of s_{max}:

$$f_{oco} = \frac{s_{max}}{\lambda} \qquad [\text{cycles / mr}] \tag{2a}$$

$$f_{oco} = \frac{s_{max}}{10^{-3}\,\lambda\ \text{fl}} \qquad [\text{cycles / mm}] \tag{2b}$$

In equation 2, s_{max} and fl are expressed in mm and the wavelength λ is expressed in micrometers. The dimensionless normalized frequency f_n is defined by:

$$f_n = \frac{f}{f_{oco}} \tag{3}$$

In equation 3, the units of f are the same as the units of f_{oco} . The user defined *Mathematica* functions **MTFNormalizedEqHor** and **MTFNormalizedEqVer** given in the Appendix allow for the computer calculation of analytical MTF functions. An important argument of these functions is the inequality that defines the entrance pupil. To have these functions properly output a normalized frequency the entrance pupil shape should satisfy the following rules:

1. The entrance pupil shape is input in units of mm
2. When using **MTFNormalizedEqHor** the entrance pupil shape is normalized so that s_{max} is one in the horizontal direction.
3. When using **MTFNormalizedEqVer** the entrance pupil shape is normalized so that s_{max} is one in the vertical direction.

3. EXAMPLES

Circular Entrance Pupil.



Figure 2. Circular entrance pupil with diameter D_0.

We use **MTFNormalizedEqHor** and choose a diameter of one in the inequality so as to make s_{max} equal to one as required by rule 2 above.

```
MTFNormalizedEqHor[x^2 + y^2 < (1/2)^2, {x, y}, fn] // Simplify
```

When the above expression is input, *Mathematica*, outputs the following expression for the MTF in the horizontal direction.

$$\begin{cases} 1 & \text{fn} == 0 \\ \frac{-2\,\text{fn}\sqrt{1-\text{fn}^2}+2\,\text{ArcCos}[\text{fn}]}{\pi} & 0 < \text{fn} < 1 \end{cases} \tag{4}$$

Using equation 2a the optical cutoff frequency f_{oco} associated with equation (4) is D_0/λ where D_0 is the entrance pupil diameter in mm. The above expression agrees with equation 11-4 in reference 6. Because of the symmetry of the circular pupil, equation 4 gives the incoherent diffraction MTF in either the horizontal or vertical direction.

Square Entrance Pupil.



Figure 3. Square entrance pupil with side a.

We use **MTFNormalizedEqHor** to find the equation for the MTF of a square entrance pupil and choose the length of a side to be equal to one so as to make s_{max} equal to one as required by rule 2 above.

```
MTFNormalizedEqHor[Abs[x] < 1/2 && Abs[y] < 1/2, {x, y}, fn]
```

When the above expression is input, *Mathematica*, outputs the following expression for the MTF in the horizontal direction.

$$\{\, 1 - \text{fn} \quad 0 \le \text{fn} < 1 \tag{5}$$

This agrees with equation 1.26 in reference 8. Using equation 2a, the optical cutoff frequency f_{oco} associated with equation 5 is a/λ where a is the length of a side in mm. Square symmetry implies equation 5 also applies to the vertical direction.

Semi-Circular Entrance Pupil.



Figure 4. Semi-circular entrance pupil with diameter D_0.

We use **MTFNormalizedEqHor** to find the equation for the MTF of the semi-circular pupil and choose the diameter to be equal to one so as to make s_{max} equal to one in the horizontal direction as required by rule 2 above.

```
MTFNormalizedEqHor[x^2 + y^2 < (1/2)^2 && y > 0, {x, y}, fn] // Simplify
```

When the above expression is input, *Mathematica* outputs the following expression for the MTF in the horizontal direction.

$$\begin{cases} 1 & \text{fn} == 0 \\ \dfrac{-2\,\text{fn}\sqrt{1-\text{fn}^2}+2\,\text{ArcCos}[\text{fn}]}{\pi} & 0 < \text{fn} < 1 \end{cases} \tag{6}$$

In equation 6, the optical cutoff frequency f_{oco} in units of cycles/mr is given by $_0/\lambda$.

To find the MTF in the vertical direction, use **MTFNormalizedEqVer** and choose the diameter equal to two so as to make s_{max} equal to one in the vertical direction.

```
MTFNormalizedEqVer[x^2 + y^2 < 1^2 && y > 0, {x, y}, fn]
```

When the above expression is input, *Mathematica* outputs the following expression for the MTF in the vertical direction.

$$\frac{2\left(\begin{cases} -\text{fn}\sqrt{1-\text{fn}^2}+\text{ArcCos}[\text{fn}] & 0 \leq \text{fn} < 1 \end{cases}\right)}{\pi} \tag{7}$$

In equation 7, the optical cutoff frequency f_{oco} is given by $D_0/(2\,\lambda)$.

4. VERIFICATION

Circular Entrance Pupil. Equation 4 is a well-known expression known to be true[6]. We get evidence for the correctness of **NMTFNormalizedHor** by comparing its output with the results of equation 4. The command that generated the numerical results is

```
NMTFNormalizedHor[x^2 + y^2 < (1/2)^2, {x, y}, 30.]
```

In the graph below the 30 points correspond to numerical results obtained from the above command and the solid line is computed using equation 4.



Figure 5. Comparison of numerical and analytical calculations for circular entrance pupil.

Square Entrance Pupil. Equation 5 is a well-known expression (see equation 1.26 in reference 8). We get evidence for the correctness of **NMTFNormalizedHor** by comparing its output with the results of equation 5. The command that generated the numerical results is

$$\texttt{NMTFNormalizedHor}\left[\texttt{Abs[x]} < \frac{1}{2}\ \texttt{\&\&}\ \texttt{Abs[y]} < \frac{1}{2},\ \{\texttt{x}, \texttt{y}\},\ 30.\right]$$

In the graph below the points correspond to numerical results obtained from the above command and the solid line is computed using equation 5.



Figure 6. Comparison of numerical and analytical calculations for a square entrance pupil.

Semi-Circular Entrance Pupil. We use **NMTFNormalizedHor** to gain confidence in this function and **MTFNormalizedEqHo**r by determining if they are consistent. The command used to generate the numerical results is

$$\texttt{NMTFNormalizedHor}\left[\texttt{x}^2 + \texttt{y}^2 < \left(\frac{1}{2}\right)^2\ \texttt{\&\&}\ \texttt{y} > 0,\ \{\texttt{x}, \texttt{y}\},\ 30.\right]$$

In the graph below the points correspond to numerical results obtained from the above command and the solid line is computed using equation 6.



Figure 7. Comparison of numerical and analytical calculations for a semi-circular entrance pupil.

The above calculations served to verify **MTFNormalizedEqHor** and **NMTFNormalizedEqHor**. Subsequent calculations will verify **MTFCyclesPerMmEqHor** and **NMTFCyclesPerMmHor**. When the output is given in cycles per mm, besides specifying the entrance pupil size it is necessary to also specify the wavelength of the incident radiation λ and the optical system focal length fl. In calculations where the output is in cycles per mm it is assumed that λ equals 10 micrometers and fl equals 10 mm.

Elliptical Entrance Pupil.



Figure 8. Elliptical entrance pupil with major axis of 5.0 mm and minor axis of 2.5 mm.

The following commands output the equations and numerical tables for the horizontal and vertical MTF.

$$\texttt{MTFCyclesPerMmEqHor}\left[\left(\frac{x}{1.25}\right)^2+\left(\frac{y}{2.5}\right)^2<1,\ \{x,y\},\ 10,\ 10,\ f\right]$$

$$\texttt{NMTFCyclesPerMmHor}\left[\left(\frac{x}{1.25}\right)^2+\left(\frac{y}{2.5}\right)^2<1,\ \{x,y\},\ 10,\ 10,\ 2.5,\ 30.\right]$$

$$\texttt{MTFCyclesPerMmEqVer}\left[\left(\frac{x}{1.25}\right)^2+\left(\frac{y}{2.5}\right)^2<1,\ \{x,y\},\ 10,\ 10,\ f\right]$$

$$\texttt{NMTFCyclesPerMmVer}\left[\left(\frac{x}{1.25}\right)^2+\left(\frac{y}{2.5}\right)^2<1,\ \{x,y\},\ 10,\ 10,\ 5.0,\ 30.\right]$$

Space limitations preclude exhibiting the equation and table produced by these commands. A graph comparing the analytical and numerical calculation is shown below.



Figure 9. Comparison of numerical and analytical calculations for entrance pupil of Fig. 8.

Figure 9 demonstrates agreement between analytical and numerical calculations done by *Mathematica*. CODE V calculations done on the entrance pupil of Fig. 8 are in agreement with the numerical calculations exhibited in Fig. 9.

Distributed Entrance Pupil.



Figure 10. Distributed aperture.

In Fig. 10 the center-to-center distance in the vertical and horizontal directions is 3.8 mm. Each of the four clear areas has a diameter of 1.2 mm.

By symmetry, the MTF in the vertical and horizontal directions is the same. The commands used to define the distributed aperture are given below.

$$\text{Region1} = x^2 + (y - 1.9)^2 < 0.6^2 ;$$
$$\text{Region2} = x^2 + (y + 1.9)^2 < 0.6^2;$$
$$\text{Region3} = (x - 1.9)^2 + y^2 < 0.6^2;$$
$$\text{Region4} = (x + 1.9)^2 + y^2 < 0.6^2;$$
$$\text{Region} = \text{Region1} \,||\, \text{Region2} \,||\, \text{Region3} \,||\, \text{Region4};$$

The commands used to generate the equation and table that define the horizontal or vertical MTF are given below.

```
MTFCyclesPerMmEqVer[Region, {x, y}, 10., 10., f]

NMTFCyclesPerMmVer[Region, {x, y}, 10., 10., 5, 30]
```

Space limitations preclude exhibiting the equation or table produced by the above commands. A comparison of the analytical and numerical results as generated by the above commands is shown below.



Figure 11. Comparison of numerical and analytical calculations for the distributed aperture of Fig. 10.

The calculations for the distributed aperture of Fig. 10 were repeated using CODE V and were in agreement with the numerical *Mathematica* calculations exhibited in Fig. 11.

Racetrack Aperture.



Figure 12. Racetrack aperture.

In Figure 12 a clear circular entrance pupil 5.0 mm in diameter is truncated to be 3.4 mm in the vertical direction.

The following commands output the equations and numerical tables for the horizontal and vertical MTF.

$$\texttt{MTFCyclesPerMmEqHor}\left[x^2 + y^2 < 2.5^2 \;\&\&\; \texttt{Abs}[y] < 1.7,\ \{x,\ y\},\ 10,\ 10,\ f\right]$$

$$\texttt{MTFCyclesPerMmEqVer}\left[x^2 + y^2 < 2.5^2 \;\&\&\; \texttt{Abs}[y] < 1.7,\ \{x,\ y\},\ 10,\ 10,\ f\right]$$

$$\texttt{NMTFCyclesPerMmHor}\left[x^2 + y^2 < 2.5^2 \;\&\&\; \texttt{Abs}[y] < 1.7,\ \{x,\ y\},\ 10.,\ 10.,\ 5.0,\ 25.\right]$$

$$\texttt{NMTFCyclesPerMmVer}\left[x^2 + y^2 < 2.5^2 \;\&\&\; \texttt{Abs}[y] < 1.7,\ \{x,\ y\},\ 10.,\ 10.,\ 3.4,\ 25.\right]$$

Space limitations preclude exhibiting the equations or tables. A graphic comparing the analytical and numerical calculations is shown below.



Figure 13. Comparison of numerical and analytical calculation for racetrack entrance pupil.

Figure 13 demonstrates agreement between analytical and numerical calculations done by *Mathematica*. CODE V calculations done on the entrance pupil of Fig. 12 are in agreement with the numerical calculations exhibited in Fig. 13.

5. SUMMARY AND CONCLUSIONS

In this paper, functions written in *Mathematica* have been exhibited and demonstrated that take as input the shape and size of the entrance pupil and then analytically or numerically calculate the incoherent diffraction MTF in the vertical or horizontal direction. Although CODE V, ZEMAX and other programs can numerically calculate these incoherent diffraction MTFs we are not aware of any other code that can take as input a description of the entrance pupil and output an analytical expression for the MTF in the specified direction. If an exact description of the entrance pupil is input, then the functions exhibited here will output an exact expression for the MTF in the specified direction.

The functions exhibited here have been verified with several entrance pupil shapes by comparing *Mathematica* generated analytical and numerical expressions against themselves and with numerical calculations done by CODE V. Another method for verifying the functions exhibited here is by comparing results produced by those functions with well-known results.

For a sufficiently complicated description of the entrance pupil, the functions exhibited here that produce analytical results may grind on endlessly and fail to produce an analytical result. For an entrance pupil that has sufficiently fine detail, the numerical functions exhibited here may need tweaking to properly reflect changes in the MTF caused by the fine detail.

This paper demonstrates a computer program that produces analytical MTF results for cases where it would be either too hard or too tedious to get analytical results from hand calculation and can serve as a user's manual for the functions given in the Appendix.

6. ACKNOWLEDGEMENTS

M. Friedman thanks Bill Blecha for bringing the calculation of a MTF problem he had to my attention. That was the seed that started the investigation done here. M. Friedman also thanks Joe Reynolds for encouraging and supporting this work.

APPENDIX – MTF Functions Used in this Paper and Notes on Their Use

All calculations were done in *Mathematica* 6.0.0.

MTFNormalizedEqHor outputs an MTF *equation* in units of normalized frequency **f** with the entrance pupil shape described by **Ineq**. Here **{x,y}** corresponds to the variables used in **Ineq**. Similar comments apply to **MTFNormalizedEqVer**.

```
MTFNormalizedEqHor[Ineq_, {x_, y_}, f_] := Module[{temp, A0, s},
  A0 = Integrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = 1/A0 Integrate[Boole[Ineq && (Ineq /. x → (x - s))], {x, -∞, ∞}, {y, -∞, ∞}, Assumptions → s ≥ 0];
  temp /. s → f]
```

```
MTFNormalizedEqVer[Ineq_, {x_, y_}, f_] := Module[{temp, A0, s},
  A0 = Integrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = 1/A0 Integrate[Boole[Ineq && (Ineq /. y → (y - s))], {y, -∞, ∞}, {x, -∞, ∞}, Assumptions → s ≥ 0]; temp /. s → f]
```

NMTFNormalizedHor outputs **NPoints** which *numerically* describe the Normalized MTF. The entrance pupil shape is described by **Ineq** and **{x,y}** corresponds to the variables used in **Ineq**. Similar comments apply to **MTFNormalizedVer**.

```
NMTFNormalizedHor[Ineq_, {x_, y_}, NPoints_] := Module[{A0, s},
  A0 = NIntegrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  Table[{s, 1/A0 NIntegrate[Boole[Ineq && (Ineq /. x → (x - s))], {x, -∞, ∞}, {y, -∞, ∞}]}, {s, 0, 1, 1/NPoints}] // Chop
 ]
```

```
NMTFNormalizedVer[Ineq_, {x_, y_}, NPoints_] := Module[{A0, s},
  A0 = NIntegrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  Table[{s, 1/A0 NIntegrate[Boole[Ineq && (Ineq /. y → (y - s))], {x, -∞, ∞}, {y, -∞, ∞}]}, {s, 0, 1, 1/NPoints}] // Chop
 ]
```

MTFCyclesPerMmEqHor outputs an *equation* for the horizontal MTF in units of cycles per mm. The inputs are an inequality **Ineq** that describes the entrance pupil shape, **{x, y}** the variables used in the inequality, the wavelength **λ** and the focal length **fl**. The variable used to describe frequency in the equation is **f**. Similar comments apply to **MTFCyclesPerMmEqVer**.

```
MTFCyclesPerMmEqHor[Ineq_, {x_, y_}, λ_, fl_, f_] := Module[{temp, A0, s},
  A0 = Integrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = 1/A0 Integrate[Boole[Ineq && (Ineq /. x → (x - s))], {x, -∞, ∞}, {y, -∞, ∞}, Assumptions → s ≥ 0];
  (* s is in mm, λ is in μ, fl is focal length in mm, f is in cycles/mm
   Factor of 10^-3 changes μ to mm *)
  temp /. s → 10^-3 λ fl f]
```

```
MTFCyclesPerMmEqVer[Ineq_, {x_, y_}, λ_, fl_, f_] := Module[{temp, A0, s},
  A0 = Integrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = 1/A0 Integrate[Boole[Ineq && (Ineq /. y → (y - s))], {y, -∞, ∞}, {x, -∞, ∞}, Assumptions → s ≥ 0];
  (* s is in mm, λ is in μ, fl is focal length in mm, f is in cycles/mm
   Factor of 10^-3 changes μ to mm *)
  temp /. s → 10^-3 λ fl f]
```

NMTFCyclesPerMmHor outputs **NPoints** that *numerically* describes the horizontal MTF. The entrance pupil shape is described by **Ineq** that utilizes variables **{x,y}**. The wavelength of the incident radiation, focal length of the optical system and the maximum displacement that corresponds to the cutoff frequency are given by **λ**, **fl** and **smax** . Similar comments apply to **NMTFCyclesPerMmVer**.

```
NMTFCyclesPerMmHor[Ineq_, {x_, y_}, λ_, fl_, smax_, NPoints_] := Module[{temp, A0, s},
  A0 = NIntegrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = Table[{s, 1/A0 NIntegrate[Boole[Ineq && (Ineq /. x → (x - s))], {x, -∞, ∞}, {y, -∞, ∞}]}, {s, 0, smax, smax/NPoints}] // Chop;
  (* s is in mm, λ is in μ, fl is focal length in mm, f is in cycles/mm
   Factor of 10^-3 changes μ to mm *)
  temp /. {a_, b_} → {a/(10^-3 λ fl), b}
 ]

NMTFCyclesPerMmVer[Ineq_, {x_, y_}, λ_, fl_, smax_, NPoints_] := Module[{temp, A0, s},
  A0 = NIntegrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = Table[{s, 1/A0 NIntegrate[Boole[Ineq && (Ineq /. y → (y - s))], {y, -∞, ∞}, {x, -∞, ∞} ]}, {s, 0, smax, smax/NPoints}] // Chop;
  (* s is in mm, λ is in μ, fl is focal length in mm, f is in cycles/mm
   Factor of 10^-3 changes μ to mm *)
  temp /. {a_, b_} → {a/(10^-3 λ fl), b}
 ]
```

MTFCyclesPerMrHor produces an *equation* that describes the horizontal MTF in units of cycles per mr. The entrance pupil shape is described by an inequality **Ineq** that utilizes variables **{x,y}**. The incident radiation has wavelength λ. The symbol **f** is used for frequency in the output equation. Similar comments apply to **MTFCyclesPerMrVer**.

```
MTFCyclesPerMrEqHor[Ineq_, {x_, y_}, λ_, f_] := Module[{temp, A0, s},
  A0 = Integrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = 1/A0 Integrate[Boole[Ineq && (Ineq /. x → (x - s))], {x, -∞, ∞}, {y, -∞, ∞}, Assumptions → s ≥ 0];
  (* s is in mm, λ is in μ, f is in cycles/mr *)
  temp /. s →  λ  f]

MTFCyclesPerMrEqVer[Ineq_, {x_, y_}, λ_, f_] := Module[{temp, A0, s},
  A0 = Integrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = 1/A0 Integrate[Boole[Ineq && (Ineq /. y → (y - s))], {y, -∞, ∞}, {x, -∞, ∞}, Assumptions → s ≥ 0];
  (* s is in mm, λ is in μ,  f is in cycles/mm *)
  temp /. s →  λ  f ]
```

NMTFCyclesPerMrHor outputs a table with **NPoints** that describes the horizontal MTF in cycles per mr. The entrance pupil shape is described by an inequality **Ineq** that utilizes variables **{x,y}**. The incident radiation has wavelength **λ** and **smax** is the displacement that corresponds to the cutoff frequency. Similar comments apply to **NMTFCyclesPerMrVer**.

```
NMTFCyclesPerMrHor[Ineq_, {x_, y_}, λ_, smax_, NPoints_] := Module[{temp, A0, s},
  A0 = NIntegrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = Table[{s, 1/A0 NIntegrate[Boole[Ineq && (Ineq /. x → (x - s))], {x, -∞, ∞}, {y, -∞, ∞}]}, {s, 0, smax, smax/NPoints}] // Chop;
  (* s is in mm, λ is in μ, f is in cycles/mr *)
  temp /. {a_, b_} → {a/λ, b}]
NMTFCyclesPerMrVer[Ineq_, {x_, y_}, λ_, smax_, NPoints_] := Module[{temp, A0, s},
  A0 = NIntegrate[Boole[Ineq], {x, -∞, ∞}, {y, -∞, ∞}];
  temp = Table[{s, 1/A0 NIntegrate[Boole[Ineq && (Ineq /. y → (y - s))], {y, -∞, ∞}, {x, -∞, ∞}]}, {s, 0, smax, smax/NPoints}] // Chop;
  (* s is in mm, λ is in μ,  f is in cycles/mm *)
  temp /. {a_, b_} → {a/λ, b}]
```

REFERENCES

[1] Night Vision Thermal Imaging Systems Performance Model, User's Guide and Reference Guide (Jan 2006)

[2] http://www.opticalres.com/

[3] http://www.zemax.com/

[4] http://www.wolfram.com/ *Mathematica* version 6.0.0 was used in this paper.

[5] Iizuka, K., [Engineering Optics], Springer-Verlag , Chapter 10 (1983)

[6] Holst, G.C., [Electro-Optical Imaging System Performance], SPIE Optical Engineering Press (2006)

[7] Kopeika, N.S., [A System Approach to Imaging], SPIE Optical Engineering Press (1998)

[8] Boreman, G.D, [Modulation Transfer Function in Optical and Electro-Optical Systems], SPIE Optical Engineering Press, Chapter 1, (2001)

[9] Driggers, R., G.; Cox, P; Edwards, T., [Introduction to Infrared and Electro-Optical Systems], Artech House Inc. (1999)

[10] Levi, L. [Applied Optics – A Guide to Optical System Design, Vol 1], Chapter 9, John Wiley & Sons (1968)

[11] Williams, T. L., [The Optical Transfer Function of Imaging Systems], Institute of Physics Publishing (1998)

[12] William, C. S.; Becklund, O. A., [Introduction to the Optical Transfer Function], SPIE Press, (2002)

[13] Barnes, K. R., [The Optical Transfer Function], American Elsevier Publishing Co, New York (1971).

[14] Nusssbaum, A. Phillips, R. A. [Contemporary Optics for Scientists and Engineers], Prentice-Hall, Inc. , Chapter 10, (1976)

[15] Brown, E., [Modern Optics], Chapter 9, (1965).

[16] Vizgaitis, J, "Aperture Effect On the Incoherent Modulation Transfer Function", Report, 2001

[17] Stamnes, J. J. and Heier, H., "Routine computation of diffraction OTF in complex optical systems", SPIE Vol 1319, p 625-626 (1990)

[18] Hopkins, H. H. , "The application of frequency response techniques in optics", Proceeedings of the Conference on Optical Instruments and Techniques, Chapman & Hall, pp 480-513 (1961)

[19] Macdonald, "The calculation of the optical transfer function", Optica Acta. vol 18, pp 269-290 (1971)

[20] Kidger, M. J., "The calculation of the optical transfer function", Optica Acta. vol 25, pp 665-680, (1978)

[21] Fiete, R. D.; Tantalo, T. A; Calus, J. R.; Mooney, J. A.; "Image quality of sparse-aperture designs for remote sensing", Opt. Eng. 41(8), pp 1957-1969 (2002)

[22] Dereniak, E. L.; Boreman, G. D.; [Infrared Detectors and Systems], John Wiley & Sons, Inc., pp 509 – 512 (1966)

[23] Goodman, J. W.; [Introduction to Fourier Optics], McGraw-Hill, Inc., pp 137 – 145 (1996).